Řešení příkladu 4.2 pomocí programu wxMaxima

Vyprázdnění paměti

(% i1) kill(all);

(% o0) $done$

Zadání známých předpokladů (zjednoduší to výpočet)

(% i1) assume (m > 0, vB>vA, mu>0, gg>0, vA>0,alpha<%pi/2);

(% o1) $[m > 0, vB > vA, \mu > 0, gg > 0, vA > 0, \frac{\pi}{2} > \alpha]$

Definice konstant pro finální vyčíslení výsledku (používáme až na konci výpočtu, pan Škoda na konzultaci odhadoval"μ" z tvaru součásti a vztahu pro kvadatický odpor, nicméně pro malé rychlosti se vychází pouze z lineárního modelu, kde "μ" značí viskozitu prostředí 1.8e-5 pro vzduch při 20°C násobenou obtékanou plochou 0.0035m^2)

(% i2) dano:[g=9.81, m=0.166, f=0.18, mu=1.8e-5*0.0035, T=0.38, L=0.5,alpha =float(20*%pi/180)];

(dano)
$[g = 9.81, m = 0.166, f = 0.18, \mu = 6.300000000000001 10^{-8}, T = 0.38, L = 0.5, \alpha = 0.3490658503988659]$

Definice síly Fx (to je prakticky jediné, co musíte odvodit na papír)

(% i3) Fx:-m*gg-2*mu*v;

(Fx) $-2\mu v - ggm$

Věta o změně kin. energie

$$mvdv = F_x(v)dx \xrightarrow[\text{separace}]{} \frac{mvdv}{F_x(v)} = dx \xrightarrow[\text{itegrace}]{} \int_{v_A}^{v_B} \frac{mvdv}{F_x(v)} = \int_0^L dx = L$$

(% i5) rov1:integrate(m*v/Fx,v,vA,vB)=L;

(rov1)
$$m\left(\frac{ggm\log(2\mu vB + ggm) - 2\mu vB}{4\mu^2} - \frac{ggm\log(2\mu vA + ggm) - 2\mu vA}{4\mu^2}\right) = L$$

rov1:logcontract(rov1);
/* zde pouze převádíme rozdíl logaritmů na logaritmus podílu */

(rov1) $$-\frac{-2m\mu vA + 2m\mu vB + gg\,m^2\log\left(\frac{ggm+2\mu vA}{2\mu vB+ggm}\right)}{4\mu^2} = L$$

Věta o změně hybnosti

$$\frac{d(mv)}{dt} = F_x(v) \xrightarrow[\text{separace}]{} \frac{mdv}{F_x(v)} = dt \xrightarrow[\text{itegrace}]{} \int_{v_A}^{v_B} \frac{mdv}{F_x(v)} = \int_0^T dt = T$$

(% i6) rov2:integrate(m/Fx,v,vA,vB)=T;

(rov2) $$m\left(\frac{\log\left(2\mu vA + ggm\right)}{2\mu} - \frac{\log\left(2\mu vB + ggm\right)}{2\mu}\right) = T$$

(% i7) rov2:logcontract(rov2);
/* zde pouze převádíme rozdíl logaritmů na logaritmus podílu */

(rov2) $$\frac{m\log\left(\frac{ggm+2\mu vA}{2\mu vB+ggm}\right)}{2\mu} = T$$

Z rovnice 2 vyjádříme logaritmus

(% i8) [globalsolve: true, programmode:true];
/* jedná se o zaříkadlo po kterém funkce solve pracuje dle našich představ */

(% o8) $$[\text{true}, \text{true}]$$

(% i9) logaritmus:solve([rov2],['log((2*mu*vA+gg*m)/(2*mu*vB+gg*m))]);
/* zde je v případě změny Fx nutný ruční zásah programátora musí se přepsat log(...) přesně tak, jak vyjde v rov2*/ /*!! solve vrací výsledek v poli (hranatých závorkách)!!*/

(logaritmus) $$[\log\left(\frac{ggm + 2\mu vA}{2\mu vB + ggm}\right) = \frac{2\mu T}{m}]$$

Pomocí umocnění "e" na obě strany rovnice se zbyvíme logaritmů a dostaneme první rovnici (rovA)

(% i10) rovA:ratsimp(%e^lhs(logaritmus[1]))=ratsimp(%e^rhs(logaritmus[1]));
/*ratsimp upraví výraz (zjednoduší)*/ /*za "logaritmus" je v hranaté závorce index 1, jelikož "logaritmus" je pole o jednom prvku */

(rovA) $$\frac{ggm + 2\mu vA}{2\mu vB + ggm} = e^{\frac{2\mu T}{m}}$$

V první rovnici nahradíme logaritmus výrazem z pravé strany vztahu "logaritmus", čímž dostaneme druhou rovnici(rovB)

(% i11) rovB:subst(rhs(logaritmus[1]),lhs(logaritmus[1]),rov1);
/*subst je nahrazení subst(nová hodnota,původní hodnota,kde nahrazuji)*/

(rovB) $$-\frac{2ggm\mu T-2m\mu vA+2m\mu vB}{4\mu^2}=L$$

Dodefinování konstanty gg

(% i12) gg:g*(-sin(α)+f*cos(α));

(gg) $$(\cos(\alpha)f-\sin(\alpha))\,g$$

Řešení rovnic rovA a rovB

(% i13) [globalsolve: true, programmode: true];

(% o13) $$[\mathrm{true},\mathrm{true}]$$

(% i14) linsolve([rovA,rovB],[vA,vB]);

(% o14)
$$[vA:\frac{gg\,m^2+\left(2ggm\mu T+4\mu^2L-gg\,m^2\right)\,e^{\frac{2\mu T}{m}}}{2m\mu e^{\frac{2\mu T}{m}}-2m\mu},vB:-\frac{-gg\,m^2-4\mu^2L-2ggm\mu T+gg\,m^2\,e^{\frac{2\mu T}{m}}}{2m\mu e^{\frac{2\mu T}{m}}-2m\mu}]$$

Dosazení konkrétních čísel (do odvozených rovnic dosadíme výše definované hodnoty z pole "dano")

(% i15) ev(vA,dano);

(% o15) $$0.994904980705311$$

(% i16) ev(vB,dano);

(% o16) $$1.639005315034761$$

Zkusme zanedbat odpor vzduchu mu->0 (musíme počítat s pomocí limity, jinak by docházelo k dělení 0)

(% i17) vAv:limit(vA, mu, 0);

(vAv) $$\frac{\frac{4mL}{T}-2\sin(\alpha)gmT+2\cos(\alpha)fgmT}{4m}$$

(% i18) vBv:limit(vB, mu, 0);

(vBv) $$-\frac{-\frac{2L}{T}-\sin(\alpha)gT+\cos(\alpha)fgT}{2}$$

Na výsledku se to příliš neprojeví

(% i19) ev(vAv,dano);

(% o19) 0.9935668821984106

(% i20) ev(vBv,dano);

(% o20) 1.63801206517001

Zanedbejme i tření

(% i21) vAt:limit(vAv, f, 0);

(vAt) $$-\frac{\sin(\alpha)gT^2-2L}{2T}$$

(% i22) vBt:limit(vBv, f, 0);

(vBt) $$\frac{2L+\sin(\alpha)gT^2}{2T}$$

Výsledek se již podstatně liší

(% i23) ev(vAt,dano);

(% o23) 0.6782981285394967

(% i24) ev(vBt,dano);

(% o24) 1.953280818828924

Závěr:

V realizovaném experimentu lze zanedbat vliv odporu vzduchu bez ztráty přesnosti. Vliv tření je nezanedbatelný.Bez odporu vzduchu jsou výsledné vztahy pro výpočet rychlostí v krajních polohá výrazně jednodušší.